Chapter#8 Abbreviations
Célja
Az algoritmus célja a szakdolgozatban szereplő rövidítések azonosítása és listázása.
Szabályok
1. A rövidítések általában nagybetűsek pl. FBI, KPI, CRC.
2. A rövidítések legtöbbször zárójelben is szerepelnek az első előforduláskor: pl. Common Vulnerabilities and Exposures (CVE)
3. A legtöbb rövidítés 2-6 karakter hosszú
4. Előfordulhat bennük szám, pl. B2B.
Pszeudo-algoritmus
1. Inicializálunk egy üres listát a rövidítések tárolására.
2. Végigmegyünk a dokumentum minden bekezdésén. Honnan tudjuk, mi a bekezdés?
3. Szétválasztjuk a szöveget mondatokra. Miért kell ez a lépés? A rövidítéseknek semmi köze a mondatokhoz, mint objektumokhoz…
4. Minden mondatban megkeressük a potenciális rövidítéseket:
· Ha egy szó teljesen nagybetűs (és legalább 2-6 karakter hosszú)
· VAGY? zárójelben van (pl. "(CVE)") mi a forrása a zárójelnek? Nem inkább fordítva: pl. A KSH (Központi Statisztikai Hivatal) adatai szerint…
· VAGY? megfelel a minta: "Hosszú leírás (ABC)"
· HA a rövidítés még nincs benne a ROVIDITESEK listában HOZZÁADJUK a listához
5. A találatokból töröljük az ismétlődőket
6. A listát rendezzük ábécé szerint


WORD makró példa
Sub RoviditesekFelismereseEsListazasa()
    Dim doc As Document
    Dim para As Paragraph
    Dim mondatok() As String
    Dim mondat As Variant
    Dim szavak() As String
    Dim szo As Variant
    Dim roviditesek As Object
    Set roviditesek = CreateObject("Scripting.Dictionary")
    
    Dim reZarojeles As Object
    Dim reRovidites As Object
    Dim matches As Object
    Set reZarojeles = CreateObject("VBScript.RegExp")
    Set reRovidites = CreateObject("VBScript.RegExp")
    
    ' Zárójelben lévo minták pl. (ABC)
    With reZarojeles
        .Global = True
        .Pattern = "\(([A-Z0-9]{2,6})\)"
    End With
    
    ' Kezdobetus ellenorzés (pl. "Central Processing Unit (CPU)")
    With reRovidites
        .Global = True
        .Pattern = "([A-Z][a-z]+(\s[A-Z][a-z]+)+)\s\(([A-Z0-9]{2,6})\)"
    End With
    
    Set doc = ActiveDocument
    
    ' 1–2. Bekezdésenként végigmegyünk
    For Each para In doc.Paragraphs
        mondatok = Split(para.Range.Text, ".")
        
        ' 3. Minden mondat feldolgozása
        For Each mondat In mondatok
            szavak = Split(mondat, " ")
            
            ' 4/a. Teljesen nagybetus szavak (2–6 hosszú, betu/szám)
            For Each szo In szavak
                szo = Trim(szo)
                szo = Replace(szo, ".", "")
                szo = Replace(szo, ",", "")
                szo = Replace(szo, ";", "")
                szo = Replace(szo, ":", "")
                szo = Replace(szo, "(", "")
                szo = Replace(szo, ")", "")
                
                If Len(szo) >= 2 And Len(szo) <= 6 Then
                    If szo = UCase(szo) And szo Like "[A-Z0-9]*" Then
                        If Not roviditesek.Exists(szo) Then
                            roviditesek.Add szo, 1
                        End If
                    End If
                End If
            Next
            
            ' 4/b. Zárójelezett rövidítések keresése
            If reZarojeles.Test(mondat) Then
                Set matches = reZarojeles.Execute(mondat)
                For Each Match In matches
                    If Not roviditesek.Exists(Match.SubMatches(0)) Then
                        roviditesek.Add Match.SubMatches(0), 1
                    End If
                Next
            End If
            
            ' 4/c. Leírás + rövidítés minta felismerése
            If reRovidites.Test(mondat) Then
                Set matches = reRovidites.Execute(mondat)
                For Each Match In matches
                    Dim rovidites As String
                    rovidites = Match.SubMatches(2)
                    If Not roviditesek.Exists(rovidites) Then
                        roviditesek.Add rovidites, 1
                    End If
                Next
            End If
        Next
    Next

    ' 5. Duplikátumok már ki vannak szurve a Dictionary által
    ' 6. Ábécé sorrendbe rendezés
    Dim kulcsok As Variant
    kulcsok = roviditesek.Keys
    Call BubbleSort(kulcsok)

    ' Eredmény kiírása új dokumentumba táblázatként
    Dim outDoc As Document
    Set outDoc = Documents.Add
    outDoc.Content.InsertAfter "Talált rövidítések:" & vbCrLf & vbCrLf
    Dim tbl As Table
    Set tbl = outDoc.Tables.Add(outDoc.Range, UBound(kulcsok) + 1, 1)
    
    Dim i As Integer
    For i = 0 To UBound(kulcsok)
        tbl.Cell(i + 1, 1).Range.Text = kulcsok(i)
    Next

    MsgBox "A rövidítések listája elkészült!", vbInformation
End Sub

' Egyszeru Bubble Sort szövegek rendezéséhez
Sub BubbleSort(arr As Variant)
    Dim i As Long, j As Long
    Dim temp As String
    For i = LBound(arr) To UBound(arr) - 1
        For j = i + 1 To UBound(arr)
            If arr(i) > arr(j) Then
                temp = arr(i)
                arr(i) = arr(j)
                arr(j) = temp
            End If
        Next j
    Next i
End Sub

